
I General purpose language developed by INRIA since 1990. . .
I . . . and now widely used by industrials (Airbus, ANSSI, CEA,

Be-Sport, Bloomberg, Facebook, Jane Street Capital, Tezos,
. . . )

I http://ocaml.org
I A book is downloadable at https://realworldocaml.org
I This lesson covers only the functional part (chap 1-7)
I Online simple tutorial at http://try.ocamlpro.com

http://ocaml.org
https://realworldocaml.org
http://try.ocamlpro.com

http://ocaml.org/learn/companies.html

http://ocaml.org/learn/companies.html

λ-calcul ⊂ OCaml 6 / 33

I λ-calcul
I Variables are strings beginning_by_lowercase_letter
I Application of a to b is just a b
I (Anonymous) function function x -> body

I Some syntactic sugar
I A function with several arguments fun x y -> body

I equivalent to function x -> function y -> body
I Naming a value let x = value in body

I note that the scope of x is explicit (here body)
I equivalent to (function x -> body) value

I Naming a function let f x y = value in body
I equivalent to let f = fun x y -> value in body

I Sequencing a ; b
I equivalent to let _ = a in b



Evaluation order 7 / 33

I Computation use call by value
I evaluate arguments before application
B evaluation order is undefined between the arguments of a

function
I computing f a1 a2 computes

1. f1, a1 and a2 in a unspecified order
2. computes the call

I if you need a specific order, use let:

let f = ... in
let a1 = ... in
let a2 = ... in
f a1 a2

1the function may be any expression



Toplevel expressions and execution 8 / 33

I A toplevel expression is an expr. not contained in a larger expr.
I A toplevel naming expression (let) without a scope (no in)

I has its scope extended to all the following toplevel expr.
I provides a kind of global naming

I There is two ways of executing OCaml programs
1. using a(n interactive) REPL2 (an interpreter), utop or ocaml

it reads an expr., evaluates it and then prints the result
2. using a compiler and then executing the produced executable file

I In a REPL, you enter an expr. and terminates it by ;;

I The compilers consider that two expr. separated by a (blank)
line are toplevel expr. in sequence

2Read-eval-print-loop



Practicing 9 / 33

I The identity function

let id = function x -> x or let id x = x

I A function applying a function to a value

let eval = function f -> function x -> f x or
let eval f = function x -> f x or let eval f x = f x

I A function composing two functions

let compose f g x = f (g x) or
let compose f g = function x -> f (g x)
Partial application consists in providing less arguments than
the expected ones

Typing 10 / 33

I OCaml is a typed language with
I primitive types

I bool, int, float, char, string, ...
I type constructors (build new types from existing types)

I t list for lists of type t, t1 * t2 for pairs of type t1 and t2, ...
I t1 -> t2 for functions from t1 to t2

I Typing is static: typing correctness is checked before execution
I Types of expressions are inferred (computed) by the compiler
⇒ The programmer is not required to give them!
I The REPL prints them for toplevel expr.

# let add i = i + 10 ;;
val add : int -> int = <fun>

# add 12 ;;
- : int = 22

expressions

types
values



Typing is strict 11 / 33

I Typing is strict, each expression must be correctly typed
# add "tutu" ;;
Error: This expression has type string but an
expression was expected of type int

I There is no automatic conversion
# add 12.1 ;;
Error: This expression has type float but an expression
was expected of type int

# add (int_of_float 12.1) ;;
- : int = 22

I OCaml has no overloading: a name has only one type
# let pi = 4.0 * atan 1.0 ;;
Error: This expression has type float but an expression
was expected of type int

# let pi = 4.0 *. atan 1.0 ;;
val pi : float = 3.14159265358979312



Polymorphism 12 / 33

I What is the type of the identity function? let id x = x

I id can take any value as argument and returns this value
⇒ Types may contain type variables 'a, 'b, ...
I The type of id is ∀'a.'a -> 'a

I This is called (universal) polymorphism

# let id x = x ;;
val id : 'a -> 'a = <fun>

# let eval f x = f x ;;
val eval : ('a -> 'b) -> 'a -> 'b = <fun>

# let compose f g x = f (g x) ;;
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>

the universal quantification is implicit

?

?

Primitive data types 13 / 33

I () the nothing value of type unit

I false and true of type bool
I logical operators: not, &&, ||, ...
I comparison operators: =, <>, <, >, <=, >=

I integers of type int
I usual operators: +,-,*, /, mod, int_of_float, ...

I floating number of type float
I usual operators: +.,-.,*., /., **, float_of_int, ...

I 'a', '\n', ... of type char

I "\ta string\n" of type string
I concatenation by ^
I conversion of the primitive data types by string_of_type
I char at position i by str.[i]



Type constructors: product 14 / 33

I Tuples (e1,...,en) of type t1 * ... * tn
I no function to decompose (see later pattern matching)
I pairs when n = 2, decompose using fst and snd

I If a vector is represented by a pair, compute its norm

# let square x = x *. x ;;
val square : float -> float = <fun>

# let norm c = sqrt (square(fst c) +. square(snd c)) ;;
val norm : float * float -> float = <fun>

# norm (2.0,-1.0) ;;
- : float = 2.23606797749979

I A function applying two functions to a pair

# let apply f g c = f (fst c), g (snd c) ;;
val apply : ('a -> 'b) -> ('c -> 'd) -> 'a * 'c -> 'b * 'd = <fun>

# apply (fun x -> x + 1) (fun x -> x - 1) (4,4) ;;
- : int * int = (5,3)

Patterns 15 / 33

I A pattern is an expression made of
I value constructors and values
I variables (only one occurrence for each variable)
I holes: _

(1,true), 1, (1,_,x) are patterns
I A pattern may match a value

I if they have the same (constructor) structure
I variables and holes match any value

(1,_,x) matches (1,"er",'a') but neither 1 nor (2,"er",'a')
I When a pattern matches a value, its variables are bound to the

corresponding parts of the value
when (1,_,x) matches (1,"er",'a') it binds x to 'a'

I The let syntax is let pattern = expression [in expression]



Pattern matching 16 / 33

I Main control structure, used to decompose values
I A pattern matching case is a pattern and an expr.

I when "applied" to a value, it can succeed or fail
I if it succeeds, expr. is evaluated with the variables bound
I syntax: pattern -> expression

I A pattern matching is a sequence of pattern matching cases
I when "applied" to a value, it uses the first case to try to match
I if it fails, the next case is used
I and so on, until one case matches
I if none of the cases matches, there is a Match_failure exception
I function p1 -> e1 | . . . | pn -> en

3

I match e with p1 -> e1 | . . . | pn -> en
I equivalent to (function p1 -> e1 | . . . | pn -> en) e

I If a case is useless or is missing, the typer will raise a warning

3fun can only have one case



Practice of pattern matching 17 / 33

I Compute a⇒ b for a pair of boolean (a, b)

# let imply v = match v with
| (true ,true) -> true | (true ,false) -> false
| (false,true) -> true | (false,false) -> true ;;

val imply : bool * bool -> bool = <fun>
or
# let imply = function
| (true ,x) -> x
| (false,_) -> true ;;

val imply : bool * bool -> bool = <fun>

I A function testing if an integer is zero

# let is_zero = function
| 0 -> true
| _ -> false ;;

val is_zero : int -> bool = <fun>

I Lists are built from
I the empty list: []
I an element e and a list l : e::l

I e is the head of e::l
I l is the tail of e::l

I Lists are monomorphic
I all elements in a list must have the same type
I a list of elements of type t is of type t list

I Syntactic sugar: [e1;. . .;en] is equivalent to e1::. . .::en::[]
I Utility functions for lists

I concatenation @
I lots of utility functions in library List (head by hd, tail by tl, ...)



Recursive functions 19 / 33

I In functional languages, iteration is done by recursion
I Recursive functions are defined by let rec

I Recursion is often combined with pattern matching
# let rec insert elt = function
| [] -> [elt]
| h::t when elt <= h -> elt::h::t
| h::t -> h::(insert elt t) ;;

val insert : 'a -> 'a list -> 'a list = <fun>

I Computing Fibonacci numbers

# let rec fib = function
| n when n < 0 -> failwith "error"
| 0 -> 0
| 1 -> 1
| n -> fib (n-1) + fib (n-2) ;;

val fib : int -> int = <fun>

Functions are values 20 / 33

I Functions can
I take functions as arguments
I return functions
I can be applied partially

I For example, applying a function on all elements of a list
# let rec iter f = function
| [] -> ()
| h::t -> f h; iter f t ;;
val iter : ('a -> 'b) -> 'a list -> unit = <fun>

# iter print_string ["a";"b";"c";"n"] ;;
abcn
- : unit = ()

# let print_list = iter print_string ;;
val print_list : string list -> unit = <fun>

Another product type constructors: records 22 / 33

I Records: product types naming the sub-elements
# type ratio = { num : int ; den : int } ;;
type ratio = { num : int; den : int; }

I A value of type ratio can be defined by
# let r1 = { num = 1 ; den = 16 } ;;
val r1 : ratio = {num = 1; den = 16}

# let r2 = { r1 with num = 3 } ;;
val r2 : ratio = {num = 3; den = 16}
the order in which the fields are given is unimportant

I The field value are accessed by their name
# let add r1 r2 = {
num = r1.num * r2.den + r2.num * r1.den ;
den = r1.den * r2.den } ;;

val add : ratio -> ratio -> ratio = <fun>



Sum types 23 / 33

I Enumerations
# type dir = North | South | East | West ;;
type dir = North | South | East | West

I Generalized by variants
# type number = Int of int | Float of float | Error ;;
type number = Int of int | Float of float | Error

I Int 8, Float 5.4 and Error are of type number

# (Int 8, Float 5.4, Error) ;;
- : number * number * number = (Int 8, Float 5.4, Error)

I Values of variant types are manipulated by pattern matching
# let print_number = function
| Int n -> print_int n
| Float f -> print_float f
| Error -> print_string "error" ;;

val print_number : number -> unit = <fun>



Sum types II 24 / 33

I Sum types can be parameterized
# type 'a option = None | Some of 'a ;;
type 'a option = None | Some of ’a

I Sum types can be recursive
# type 'a list = [] | :: of 'a * 'a list ;;
type 'a list = [] | :: of ’a * ’a list

I Both the option and list types are already defined in OCaml
B In fact in OCaml, the variant constructors must be

1. a capitalized identifier
2. []
3. ::4, it will be treated as a binary infix constructor

4to be correct the type declaration should surrounds it by parenthesis

I A module is a set of type, value and function definitions5

I A module has
I a signature defining its public interface

I type definitions and type declarations for values and functions
I a structure defining its content

I any OCaml code

I Elements of the signature must be part of the structure
I In another module, one can use a public element elt of a

module M
I by M.elt
I by elt if the module was previously opened by open M

I A filename.ml file is a module Filename
I if there is a filename.mli, it provides its signature
I else everything is public (which is a bad practice)

5and modules but we won’t cover that



Abstraction 27 / 33

I One of the interest of modules is abstracting (hiding) types
I The following signature abstracts ratio

type ratio
val create : int -> int -> ratio A constructor
val add : ratio -> ratio -> ratio A manipulator
val print : ratio -> unit A destructor

I Code outside the defining module of an abstract type
I cannot use its implementation
I can only manipulate value through the offered functions

⇒ Changing the implementation of ratio does not impact clients
I Abstracting internal functions and values is also a good idea
I The signature is the ApplicationProgrammingInterface

I it generally includes constructors, manipulators and destructors for
the abstract types



Exceptions 28 / 33

I Exceptions are declared by exception

# exception Empty_list of string ;;
exception Empty_list of string

I Raised by raise

# let head = function
| [] -> raise (Empty_list "bouh!")
| hd :: tl -> hd ;;

val head : 'a list -> 'a = <fun>

I Caught by try with

try
head []

with
| Empty_list msg -> print_endline msg ;;

bouh!
- : unit = ()

Compiling and running OCaml 30 / 33

I ocaml REPL (we use utop)
I compiles and executes immediately
I prints value and types
I provides a simple line editor (bash default binding)
I #use "toto.ml";; loads and execute every expr. of toto.ml

I Two compilers exist
I a bytecode compiler ocamlc (with bytecode interpreter ocamlrun6)
I a native compiler ocamlopt that directly produces executable files

I Both are three steps compilers (XX means c or opt)
I compile signatures by ocamlXX -c YY.mli to produce YY.cmi
I compile modules by ocamlXX -c YY.ml to produce YY.cmZZ

I ZZ = o if XX = c and x if XX = opt

I linking of all the need modules by ocamlc -o WW YY1.cmZZ
YY2.cmZZ to produce the executable WW

I For a module without signature all its content is put in the cmi
6Invoking ocamlrun is optional since the bytecode file already invoke it



Automatic building ocamlbuild 31 / 33

I All files are compiled in the directory _build

I It groans if it finds compilation artefacts elsewhere!
I It has a target X.byte or X.native to indicate the compiler

I ocamlbuild -libs unix main.native
I X become the name of the executable

I Finds all the dependencies (hence the compilation order) alone
I Can run if you add -- followed by the command line args.

ocamlbuild main.byte -- file.txt

I Configuration file _tags for a finer control of build
I https://github.com/ocaml/ocamlbuild/blob/master/manual/manual.adoc

https://github.com/ocaml/ocamlbuild/blob/master/manual/manual.adoc

I Short introduction to the functional core of OCaml
« Computing values not modifying variables »

I It is our objective during this module
I We ignore

I imperative features
I objects
I first class modules, ...

I You need to practice...
I http://ocaml.org

I https://realworldocaml.org (chap 1-7 and 16)

http://ocaml.org
https://realworldocaml.org

